

Schon mal eine Präsentation mit SAS programmiert?

Sven Wichmann
HMS Analytical Software GmbH
Rohrbacher Straße 26
69115 Heidelberg
sven.wichmann@analytical-software.de

Zusammenfassung

Eine PowerPoint Präsentation mit SAS? Seit SAS Version 9.4 ist dies möglich. Dieser Beitrag stellt die Funktionalität von ODS POWERPOINT vor. Zudem sollen Einsatzmöglichkeiten sowie Vor- und Nachteile gegenüber der klassischen Präsentationserstellung mit Microsoft PowerPoint beleuchtet werden. Ebenfalls werden Lösungen zum effektiven Zusammenspiel beider Varianten kurz dargestellt.

Schlüsselwörter: PowerPoint, ODS, ODS POWERPOINT, PROC ODSTEXT, PROC ODSLIS

1 Einleitung

Mit SAS 9.4 wurde das SAS Output Delivery System (ODS) um neue Ausgabeformate erweitert. Eines dieser neuen Ausgabeformate ist Microsoft PowerPoint. Je nach beruflichem Aufgabengebiet kann es vorkommen auch PowerPoint Präsentationen erstellen zu müssen. Wie man das auch mit SAS machen kann, soll hier gezeigt werden. Zudem wird gezeigt, wie man die Outputs am besten benutzen kann.

2 Grundlagen

Das Grundgerüst für die Erstellung von PowerPoint Folien ist analog zur Erzeugung anderer Ausgabeformate mit SAS. Ohne weitere Parameter lässt sich schon mit sehr wenig Code eine einfache PowerPoint Präsentation erstellen. Eine einfache PowerPoint Datei, die ohne weitere Parameter erstellt worden ist, ist in Abbildung 1 zu sehen.

Der SAS-Code beschränkt sich auf wenige Zeilen:

```
1. ods html close;
2. ods powerpoint file="C:\KSFE2017\simple.pptx";
3. proc print
   data=sashelp.cars(obs=5);
   run;
4. ods _all_ close;
```

Zeile 1 schließt die ODS Standardausgabe im html-Format. Die zweite Code Zeile öffnet das Ausgabeziel von SAS nach PowerPoint und spezifiziert als Zielfile die hinter `file=` angegebene Datei. Zeile drei ist eine einfache Prozedur, welche Ausgaben erzeugt.

Die Ausgaben dieser Prozedur werden in die Datei „simple.pptx“ geschrieben, die zuvor als Ziel angegeben wurde. Dabei füllt ODS die zu erzeugenden Folien mit Inhalt und benutzt dabei eine Stilvorlage namens „PowerPointLight“. In der Stilvorlage sind die grundlegenden

Folienparameter, wie Schriftgröße, festgelegt. Passt der darzustellende Inhalt nicht auf eine Folie, wird der Inhalt automatisch auf mehrere Folien verteilt. Jeder Aufruf einer weiteren ausgabeerzeugenden Prozedur erzeugt eine oder mehrere zusätzliche Folien. Die Datei enthält, abgesehen von dem darzustellenden Inhalt, nur ein Datum und eine Seitenzahl in der Fußnote.

Werden Titel und Fußnoten benutzt, kommt man dem gängigen PowerPoint Format schon sehr nahe.

In Abbildung 2 wurden über die SAS Statements `title1` und `title2` zwei Titel vergeben. Des Weiteren wurden über `footnote1` und `footnote2` zwei Fußnoten hinzugefügt. Diese „Basis“-Folien bilden eine gute Grundlage, um sie in andere PowerPoint Präsentationen zu importieren.

Mehr als zwei Titel und Fußnoten sollten allerdings nicht vergeben werden. Da dies nicht vorgesehen ist, kann es zu ungewünschten Ergebnissen kommen.

Des Weiteren sollte die bei PowerPoint Folien nötige recht große Schriftart beachtet werden. Sollen, wie in den bisher gezeigten Beispielen, Tabellen präsentiert werden, sollte die darzustellende Tabelle, abhängig vom Inhalt, nicht mehr als 4 bis 6 Spalten und nicht mehr als 5 bis 7 Zeilen enthalten. In den bisherigen Beispielen wurde daher die `obs=5` Option im `set`-Statement verwendet.

Obs	Make	Model	Type	Horsepower	LPHK
1	Ford	Excursion 6.8 XLT	SUV	310	23.5000
2	Hummer	H2	SUV	316	23.5000
3	Dodge	Viper SRT-10 convertible 2dr	Sports	500	19.5833
4	Land Rover	Range Rover HSE	SUV	282	19.5833
5	Land Rover	Discovery SE	SUV	217	19.5833

Die Daten stammen aus Zeilen vor der Diselgate Affäre.
LPHK=Liter/100km. Umgerechnet von Meilen pro Gallone (MPG): 1/(0.42*MPG*100)

Abbildung 1: Einfache ODS PowerPoint-Ausgabe

Die 5 größten Spritverbraucher

Sortiert nach Verbrauch in Liter pro 100 Kilometer

Obs	Make	Model	Type	Horsepower	LPHK
1	Ford	Excursion 6.8 XLT	SUV	310	23.5000
2	Hummer	H2	SUV	316	23.5000
3	Dodge	Viper SRT-10 convertible 2dr	Sports	500	19.5833
4	Land Rover	Range Rover HSE	SUV	282	19.5833
5	Land Rover	Discovery SE	SUV	217	19.5833

Die Daten stammen aus Zeilen vor der Diselgate Affäre.
LPHK=Liter/100km. Umgerechnet von Meilen pro US Gallone (MPG): 235/MPG

Abbildung 2: Folie mit Titeln und Fußnoten

2.1 Datum und Uhrzeit

Wie oben bereits beschrieben, enthalten die Folien standardmäßig das aktuelle Datum und die Seitenzahl. Um die Ausgabe des Datums in der Fußzeile komplett abzuschalten, kann die SAS-Option „NODATE“ verwendet werden. Analog dazu schaltet die SAS-Option „NONUMBER“ die Ausgabe der Seitenzahl ab.

Ohne weitere Einstellungen im ODS Statement sind sowohl das Datum als auch die Seitenzahl dynamisch. Das bedeutet, dass sobald die aus SAS erstellte Präsentation geöffnet wird, sich das Datum im Folienfooter ändert. Soll das dynamische Ändern des Datums verhindert werden, kann die Option „SASDATE“ an das ODS PowerPoint Statement mit übergeben werden. Der Aufruf würde sich dadurch wie folgt ändern:

```
ods powerpoint file="C:\KSFE2017\simple.pptx" SASDATE;
```

Damit erstellt ODS PowerPoint weiterhin ein Datum in der Folienfußnote. Dieses Datum wird sich jedoch nicht ändern, wenn die Präsentation geöffnet wird. Die Seitenzahl dagegen bleibt immer dynamisch.

2.2 Bullets, Listen und Texte

Geht es bei der Präsentation um die Darstellung von Texten, werden diese üblicher Weise in Form von Aufzählungen oder Stichpunkten abgebildet. SAS liefert zu diesem Zweck die Prozeduren `PROC ODSTEXT` und `PROC ODSLIS` mit.

`PROC ODSTEXT` wird im Zusammenhang mit ODS PowerPoint dazu verwendet, um Textblöcke zu erstellen. Die einzelnen Textabschnitte werden dabei mit einem `p`-Statement innerhalb von `ODSTEXT` definiert. Weiterhin kann man über das Element `ITEM` mit Hilfe von `ODSTEXT` Listen erstellen lassen. Analog dazu erstellt `PROC ODSLIS` über `p`-Elemente innerhalb eines `ITEM` Elements eine Reihe von Listen. Als Listenelement werden, ohne weitere Anpassungen, Bulletpoints verwendet.

`ODSTEXT` kann ebenfalls Listen erstellen wie es mit `ODSLIS` möglich ist. Allerdings bietet `ODSTEXT` darüber hinaus weitere Gestaltungsmöglichkeiten. So ist es zum Beispiel möglich Text und Bulletpointlisten in einem `ODSLIS` Aufruf zu erstellen. Mit beiden Prozeduren kann sowohl feststehender Text erstellt werden, als auch Inhalte aus SAS Datensätzen gelistet werden.

Das Beispiel in Abbildung 3 zeigt die Verwendung mehrerer ODS TEXT Elemente, die über ODS LAYOUT GRIDDED (siehe Kapitel 3.5.2) arrangiert wurden. Die Texte in schwarz (Titel), rot (linker Block) und grün (rechter Block) sind feste Texte, die über p-Statements erzeugt wurden. Der Text in blau (unterer Block) listet Daten aus einem Datensatz, der über ein data= Statement eingelesen wurde. Dabei wird jede Zeile als Aufzählungspunkt dargestellt.




Abbildung 3: ODS TEXT Beispiele

3 Mehr Gestaltungsmöglichkeiten

Mit der neuen ODS PowerPoint Ausgabe lassen sich aber auch direkt ansehnliche Folien erstellen. Auch wenn es zurzeit keine Möglichkeit gibt, ein vorhandenes Template als Basis für die Ausgaben aus SAS zu benutzen, so gibt es jedoch einige andere Gestaltungsmöglichkeiten und Wege wie dieses Manko etwas gemildert werden kann. Die Optionen, die SAS zu Verfügung stellt, können entweder als eigenständige ODS PowerPoint Statements abgesetzt oder mit anderen z.B. dem file=-Statement zusammen kombiniert werden.

3.1 Folienhintergrund

Die einfachste Möglichkeit Folien aufzuwerten, ist die Verwendung von farbigen Hintergründen. Über die Option backgroundcolor= bietet SAS die Möglichkeit einen farbigen Folienhintergrund zu erstellen. Die Farben müssen in Anführungszeichen übergeben werden. Dabei können die üblichen Farbnamen verwendet werden (z.B. green, blue, black, yellow, ...), oder aber die hexadezimalen RGB Farbwerte in Form von „#RRGGBB“.

Der folgende Aufruf erzeugt Folien mit mintgrünem Hintergrund:
`ods powerpoint options(backgroundcolor="#66ff99");`

Soll der Folienhintergrund noch etwas mehr Abwechslung bieten als nur eine einfache Farbe, so kann die Option backgroundimage= verwendet werden. Damit lassen sich nicht nur vorhanden Bilder verwenden (siehe nächstes Kapitel), sondern auch farbige Hintergründe mit Verlauf. Die Farbverläufe werden dem CSS3 Standard¹ folgend definiert.

¹ <https://www.w3.org/TR/css3-images/>

Der folgende Code erzeugt einen kreisförmigen Farbverlauf, beginnend im Zentrum, von gelb nach blau. Das Ergebnis ist in Abbildung 4 zu sehen.

```
ods powerpoint
  options(backgroundimage=
"radial-
gradient(center,yellow,blue)");
```

Als Farbverlaufstypen stehen `radial-gradient` und `linear-gradient` zur Verfügung. Über weitere Parameter bei den Farbverläufen lassen sich nahezu beliebig komplexe Farbverläufe erzeugen.



Abbildung 4: Folienhintergrund mit Farbverlauf

3.1 Hintergrundbilder

Eine weitere Möglichkeit, die erstellten Präsentationen aufzuwerten, bietet die Verwendung von Hintergrundbildern. Das entsprechende Kommando dazu ist `options(backgroundimage="<PFAD_ZUM_BILD>")`. Als eigenständiges ODS Statement sieht es wie folgt aus:

```
ods PowerPoint options(backgroundimage="C:\ bild.jpg");
```

Auf diese Art und Weise lassen sich unterschiedliche Folien mit verschiedenen Hintergrundbildern versehen.

3.2 „Corporate identity“

Muss oder möchte man bei der Erstellung der Präsentation ein firmenspezifisches Design verwenden, kann dies über die Verwendung von Hintergrundbildern geschehen. PowerPoint erlaubt es, Folien und ein vorhandenes Corporate Design als Bild zu exportieren. Allerdings werden diese Bilder nur mit einer niedrigen Auflösung von 96dpi exportiert, was bei einer 4:3 Standard-Präsentation einem exportierten Bild mit 960 x 720 Pixeln entspricht. Diese recht kleinen Bilder wirken als Präsentationshintergrund gegebenenfalls unscharf. Auf den Microsoft Support Seiten² findet sich eine Anleitung, wie Folien aus PowerPoint exportiert werden können und wie die Qualität der exportierten Dateien angepasst werden kann.

Durch die Verwendung mehrerer `backgroundimage` Optionen lassen sich dann die verschiedenen Vorlagen für unterschiedliche Folien in SAS verwenden.

² <https://support.microsoft.com/de-de/help/827745/how-to-change-the-export-resolution-of-a-powerpoint-slide>

3.3 Folienformat

Klassische PowerPoint-Folien liegen im 4:3 Format vor. Dies ist auch der Standard bei der Ausgabe von Folien aus SAS. Wird dagegen auf modernen Ausgabegeräten präsentiert, sollten die Folien dementsprechend im 16:9 Format vorliegen. Um das Präsentationsformat zu ändern, muss die SAS Option `papersize` entsprechend geändert werden.

- 16:9 Format: `options papersize=(240mm 143mm);`
- 4:3 Format: `options papersize=(240mm 190.5mm);`

Wird das Folienformat auf 16:9 geändert, muss allerdings beachtet werden, dass in der Höhe weniger Platz für die Inhalte verfügbar ist. Bei Tabellen gehen, je nach Inhalt, zwei bis drei Zeilen verloren. Die Anzahl der Spalten kann dagegen um ein bis zwei höher ausfallen. Da SAS, wie eingangs beschrieben, die Inhalte an den verfügbaren Platz anpasst, können exportierte Grafiken unter Umständen in einer 16:9 Präsentation zu klein werden.

3.4 Style Templates

In SAS stehen für die Erstellung von PowerPoint Folien die zwei Style-Templates `PowerPointLight` und `PowerPointDark` zur Verfügung. Wird kein „Style“ angegeben, so wird per default `PowerPointLight` verwendet.

Sollen eigene Stilvorlagen Anwendung finden, so empfiehlt SAS dringend die beiden vorhandenen Vorlagen zu verwenden und wo nötig zu modifizieren. Andere Stilvorlagen können unter Umständen zu unbefriedigenden Ergebnissen führen, da eine reibungslose Verwendung mit ODS PowerPoint nicht garantiert werden kann. Welche Elemente sich in den vorhandenen Stilvorlagen anpassen lassen, zeigt `PROC TEMPLATE`.

Der Aufruf:

```
proc template;  
  source styles.powerpointlight;  
run;
```

listet im SAS-Log alle vorhandenen Elemente der Stilvorlage `PowerPointLight` auf. Aufbauend auf den vorhandenen Stilvorlagen lassen sich viele Parameter in der zu erstellenden PowerPoint Präsentation anpassen. Auf diese Weise lassen sich schnell Präsentationslayouts erstellen, die den Firmenstandards genügen oder den eigenen Farb- oder Designvorlieben entsprechen.

3.5 Folienlayout

Für die Gestaltung der Folieninhalte verwendet ODS PowerPoint sogenannte Layouts. Dabei lassen sich zwei verschiedene Typen von Layouts unterscheiden. Zum einem bringt SAS vordefinierte Layouts mit, zum anderen können auch eigene Layouts definiert werden.

3.5.1 Vordefinierte Layouts

ODS PowerPoint unterstützt Gitter-Layouts und liefert die drei PowerPoint typischen Layouts gleich als Parameter mit:

- `LAYOUT=TitleSlide` → Titel-Folie
- `LAYOUT=TitleandContent` → Folie mit Titel und Inhalt
- `LAYOUT=TwoContent` → Zwei Spalten mit Inhalt.

Das Layout „TitleSlide“ erzeugt eine Folie, wie sie in Abbildung 5 zu sehen ist. Die Titel werden über `p`-Statements innerhalb von `ODSTEXT` erstellt. Dabei ist zu beachten, dass nicht mehr als zwei Titel unterstützt werden. Alle Titel, die über `title`-Statements erstellt wurden, werden ignoriert. Für die Titelfolie kann eine Fußnote vergeben werden (über ein `footnote`-Statement). Jede weitere Fußnote wird ignoriert.



Abbildung 5: Titelfolie erstellt mit `Layout=TitleSlide`

Das vordefinierte Folienlayout „TitleAndContent“ erzeugt Folien, wie sie in Abbildung 6 zu sehen ist. Im Unterschied zum Titel-Layout können die Titel hier über SAS `title1`- und `title2`-Statements erstellt werden. Folien, die mit dem Layout „TitleAndContent“ erstellt werden, sind Folien ohne Layout recht ähnlich. Im Unterschied hierzu werden jedoch keine neuen Folien erzeugt, wenn eine neue SAS-Prozedur aufgerufen wird. Erst wenn der Inhalt nicht mehr auf die Folie passt, wird eine weitere Folie erstellt. Für Slides mit dem Layout „TitleAndContent“ können zwei Titel und zwei Fußnoten vergeben werden. Jeder weitere Titel oder jede weitere Fußnote wird ignoriert. Das Layout gilt ab der Definition für jede weitere Folie, bis entweder ein neues Layout spezifiziert, oder über ein „`LAYOUT=_NULL_`“ das Layout zurückgesetzt wird.



Abbildung 6: Folie für Inhalte erstellt mit `Layout=TitleAndContent`

Das dritte Standardlayout ist „TwoContent“. Ein Beispiel ist in Abbildung 7 zu sehen. Dieses erzeugt ein zweiseitiges Layout. Dabei wird der Inhalt der ersten ausgabeerzeugenden Prozedur in die linke Hälfte geschrieben und die Ausgaben der zweiten Prozedur in die rechte Hälfte.

Marke	LPHK	Marke	LPHK
Hummer	23.5	Scion	7.5
Land Rover	17.4	MINI	8.9
GMC	15.5	Honda	9.5
Isuzu	14.7	Saturn	9.7
Cadillac	14.4	Hyundai	10.7

Abbildung 7: Folie mit zweiseitigem Inhalt erzeugt mit `Layout=TwoContent`

Sind die Ausgaben der ersten Prozedur größer als auf eine Folie passt, wird eine zweite Folie erstellt und nicht in die rechte Hälfte umgebrochen. Für Titel und Fußnote gilt das gleiche wie beim Layout.

„TitleAndContent“. Das Layout muss ebenfalls über ein „LAYOUT=_NULL_“ Statement explizit beendet werden. Andernfalls gilt es für alle weiteren Folien.

Die folgenden Zeilen zeigen den vollständigen Code für die in den Abbildungen 5 bis 7 erzeugten Folien. Nicht dargestellt ist die Erzeugung des für die dritte Folie erforderlichen Datensatzes. Dieser enthält Daten aus dem sashelp.cars Datensatz. Die Erzeugung der Inhalte erfolgt mit der in Kapitel 2.2 vorgestellten Prozedur PROC ODSTEXT.

```
ods html close;
ods powerpoint file="C:\KSFE2017\styles1.pptx"
               layout=TitleSlide
               options(backgroundimage='C:\Bilder\light-bulb.jpg');
proc odstext;
  p "Standard ODS PowerPoint Layouts" / style=PresentationTitle;
  p "Der Titel Slide (layout=TitleSlide)" / style=PresentationTitle2;
run;

ods powerpoint layout=_null_;

*Next Slide *;
ods powerpoint layout=TitleAndContent;

title1 "Die 'Standard Folie'";
proc odstext;
  p ""; p "";
  list;
    item "Slide mit Inhalt und Titel";
    item "Aufruf im ODS Statement:";
    item "ods powerpoint layout=TitleAndContent";
  end;
run;

*Next Slide *;
ods powerpoint layout=TwoContent;

title1 "Eine Folie mit zwei Splaten";
proc report data=cars_n2(obs=5) HEADSKIP HEADLINE;
  columns make lphk;
  define make / display "Marke" width=15;
  define lphk / display "LPHK" width=15;
run;
proc report data=cars_n1(obs=5) HEADSKIP HEADLINE;
  columns make lphk;
  define make / display "Marke" width=15;
  define lphk / display "LPHK" width=15;
run;

ods powerpoint close;
```

3.5.2 Benutzerdefinierte Layouts

Wird für die Erstellung der Folien mehr Gestaltungsspielraum benötigt, als durch die drei vordefinierten Layouts möglich ist, kann über „LAYOUT=GRIDDED“ ein benutzerdefiniertes Layout erstellt werden. Damit wird es auch möglich Teile der Folie mit einem zweispaltigen Layout zu benutzen und andere Teile mit einem einspaltigem Layout. Zudem lassen sich die einzelnen Layout Elemente freier positionieren und in ihrer Größe beeinflussen.

Die in Abbildung 3 gezeigte Folie nutzt insgesamt vier Layout Elemente mit unterschiedlichen Inhalten. Die wichtigsten Codezeilen sollen im Folgenden kurz dargestellt werden.

```

1. ods html close;
2. ods powerpoint file="C:\styles1.pptx" layout=TitleSlide
   options(backgroundimage='C:\Bilder\light.jpg');
3. ods layout gridded columns=1;
4.   ods region;
5.   proc odstext;
6.     p "";
       p "PowerPoint Do's and Dont's" / style=[just=Center];
       run;
7. ods layout end;

8. ods layout gridded columns=2 rows=1 column_widths=(48% 48%);
9. ods region;
10.  proc odstext;
11.    p "Bitte nicht:";
12.      list / style=[liststyletype=decimal color=red
fontsize=20pt];
13.      item "Keine Bulletpoints";
       item "Nicht zu viel Text";
       item "Nicht langweilen";
14.    end;
       run;
15. ods region;
16.  proc odstext;
17.    p "Aber bitte:" ;
18.      list / style=[bullet=check color=green];
19.      item "Pausen machen";
       item "aber nicht zu viele";
       item "Gutes Thema";
20.    end;
       run;
21. ods layout end;
22. ods layout gridded columns=1;
23. ods region;
24.  proc odstext;
25.    p "Verbrauchsdurchschnitt - Daten bitte:";
       run;
26. ods layout end;
27. ods layout gridded columns=1;

```

```
28. ods region;
29. proc odstext data=cars_n
    (where=(make in ("Audi", "BMW", "Mercedes-Benz", "Volkswagen")));
30.   list / style=[color=blue fontsize=20pt];
31.   item "Marke: " || strip(make) || " -
Verbrauchsdurchschnitt:"
    || put(LPHK,4.2);
32.   end;
    run;
33. ods layout end;
```

Kommentare zu den einzelnen Zeilen:

- #1 Schließen der html-Ausgabe.
- #2 Öffnen der Ausgabe nach ODS PowerPoint, Angabe der Zieldatei styles1.pptx, Verwendung des Layouts TitleSlide und einbinden eines Hintergrundbildes durch die Option „backgroundimage=“.
- #3 Beginn eines neuen „Gridded“-Layouts mit einer Spalte.
- #4 Erstellung des ersten Elementes im erzeugten Layout.
- #5 Start von PROC ODSTEXT zur Erzeugung des Inhaltes im gerade erstellten Element.
- #6 Textauflistung mit einer Leerzeile, um den Abstand zum Titel zu erhöhen.
- #7 Beenden des in #3 begonnenen Layouts.
- #8 Beginn eines zweispaltigen Layouts. Die Breite der beiden Spalten wird jeweils auf 48% festgelegt.
- #9 Beginn des ersten Elementes im neu erzeugten Layout. Da in #8 ein zweispaltiges Layout erzeugt wurde, entspricht dies der linken der beiden Spalten.
- #10 Start von PROC ODSTEXT zur Erzeugung des Inhaltes im linken Layout-Element.
- #11 p-Element als Überschrift für die darauf folgende Aufzählung.
- #12 Beginn der Aufzählung im linken Elementblock. Als Attribut wird eine numerische Liste mit rotem Text erzeugt. Die Schriftgröße wird auf 20pt festgelegt.
- #13 Erstellung der Inhalte in der durchnummerierten Liste
- #14 Beenden der Listenerstellung
- #15 Beginn des zweiten Elements im zweispaltigen Layout. Dies entspricht der rechten Spalte im in #8 erzeugten zweispaltigen Layouts.
- #16 - Erzeugung der Inhalte in der rechten Spalte. Dies entspricht den Zeilen #10 bis #21.
- #21 Allerdings werden für die Liste grüner Text und Haken als Aufzählungszeichen verwendet.
- #22 - Beginn eines dritten Layout-Elementes mit einer Zeile. Dieses Element dient als Überschrift für die darauf folgenden Inhalte.
- #26
- #27- Beginn des vierten Layout-Elementes.
- #28
- #29 Beginn der Prozedur ODSTEXT. Hier wird der Prozedur über data= ein Datensatz zur Erzeugung der Inhalte mit übergeben. Das where-Statement begrenzt die Inhalte auf vier Zeilen.

- | | |
|-----|----------------------------------------------------------------------------------------------------------------------------------------------------------------|
| #30 | Beginn der Aufzählungsliste. Für die Textausgabe wird der Text in blau spezifiziert. |
| #31 | Beginn der Listenerzeugung. Die Inhalte der Liste werden mit dem Inhalt des Eingangsdatensatzes gefüllt und der Datensatzreihenfolge entsprechend dargestellt. |
| #32 | Beenden des vierten Layout-Elementes. |

Für die gezeigte Beispielfolie folgt im Anschluss an #32 ein `ods powerpoint close;` um die Erstellung der Folien abzuschließen. Sollen dagegen weitere Folien hinzugefügt werden, kann mit `ods powerpoint layout=_null_;` eine neue Folie erzeugt werden. Ansonsten füllt ODS PowerPoint die Folien selbständig mit Inhalten auf. Neue Folien werden nur dann erzeugt, wenn der Platz auf der aktuellen Folie nicht für weitere Inhalte ausreicht.

4 Abschluss

ODS PowerPoint bietet eine gute Erweiterung der SAS Ausgabeformate. Durch die Möglichkeit recht schnell und mit wenig Aufwand einfache Folien zu erstellen und diese in bestehende Vorträge einzubinden, erübrigt sich die Antwort auf die Frage, ob das PowerPoint Ausgabeformat nötig ist. Dies trifft vor allem dann zu, wenn Grafiken oder Tabellen in anderen PowerPoint-Folien benötigt werden. Der manchmal recht umständliche Umweg über Excel (im Falle von Tabellen) oder über den Export von Grafikdateien entfällt.

Seine Stärken spielt das neue Ausgabeformat auch dann aus, wenn regelmäßig gleiche oder ähnliche Präsentationen erstellt werden müssen. Sind lediglich neue Daten die Grundlage für ein Update der Präsentation, reicht dann ein Start des vorhandenen SAS Programms, um eine Präsentation basierend auf den neusten Daten zu erstellen.

Auch die vorhandenen Stilvorlagen erleichtern die Ausgabe auf einem Präsentationsmedium enorm. Eine Tabelle mit Daten, die niemand mehr lesen kann, sollte damit in Zukunft nicht mehr vorkommen.

Durch die vielfältigen Anpassungsmöglichkeiten der Stilvorlagen und die hohe Flexibilität bei den Folienlayouts, lassen sich zudem auch besondere Designwünsche erfüllen.